Exam 2 Take Home: Tea time

Do the following to submit with your exam, from section 4.3, pp. 163-167: #3-5, 11, 12, 18 abfg, 21, 22, 25, 40

For problems 3, 4, 5, and 11, write code that works for an arbitrary function, and then apply the code on those problems. Provide me with a script that allows me to (easily!:) reproduce your results.

Preliminaries:

3. Approximate the integral using Simpson's rule.

   (a) $\int_{-0.5}^{0} x \ln(x+1) dx$ [S]

   (b) $\int_1^3 \ln(x+1)\, dx$

   (c) $\int_{-0.25}^{0.25} (\cos x)^2 dx$ [A]

   (d) $\int_1^3 e^{\sin x}\, dx$

   (e) $\int_1^2 x^4\, dx$ [A]

4. Do question 3 using the Trapezoidal rule. [S] [A]
5. Do question 3 using the Midpoint rule. [S] [A]

In[463]:=

```
f[x_] := x * (x - 1) * E ^ x
mid[f_, a_, b_] := (b - a) * f[(b + a) / 2]
trap[f_, a_, b_] := (b - a) / 2 * (f[a] + f[b])
simpsons[f_, a_, b_] := 1 / 3 (2 * mid[f, a, b] + trap[f, a, b])
results[g_, a_, b_, exercise_] :=
 Module[{},
  Print[exercise];
  TableForm[{{trap[g, a, b]}, {mid[g, a, b]},
    {simpsons[g, a, b]}, {NIntegrate[g[x], {x, a, b}]}},
   TableHeadings → {{"trap", "mid", "simpson", "Mathematica"}, {"estimate"}}]
 ]
```

#3-5:

In[468]:=
```
g[x_] := x Log[x + 1]
{a, b} = {-0.5, 0};
results[g, a, b, "3-5(a)"]
```

3-5(a)

Out[470]//TableForm=

	estimate
trap	0.0866434
mid	0.0359603
simpson	0.0528546
Mathematica	0.0525698

In[471]:=
```
g[x_] := Log[x + 1]
{a, b} = {1.0, 3.0};
results[g, a, b, "3-5(b)"]
```

3-5(b)

Out[473]//TableForm=

	estimate
trap	2.07944
mid	2.19722
simpson	2.15796
Mathematica	2.15888

In[474]:=
```
g[x_] := Cos[x]^2
{a, b} = {-0.25, 0.25};
results[g, a, b, "3-5(c)"]
```

3-5(c)

Out[476]//TableForm=

	estimate
trap	0.469396
mid	0.5
simpson	0.489799
Mathematica	0.489713

In[477]:=
```
g[x_] := E^Sin[x]
{a, b} = {1.0, 3.0};
results[g, a, b, "3-5(d)"]
```

3-5(d)

Out[479]//TableForm=

	estimate
trap	3.47134
mid	4.96516
simpson	4.46722
Mathematica	4.4248

In[480]:=
```
g[x_] := x^4
{a, b} = {1.0, 2.0};
results[g, a, b, "3-5(e)"]
```

3-5(e)

Out[482]//TableForm=

	estimate
trap	8.5
mid	5.0625
simpson	6.20833
Mathematica	6.2

Exercise #11-12:

11. For the following values of f, x_0, and h, use the formula

$$f'(x_0) = \frac{f(x_0+h) - f(x_0-h)}{2h} - \frac{h^2}{6} f'''(\xi)$$

to approximate $f'(x_0)$.

(a) $f(x) = e^x$; $x_0 = 2$; $h = 0.1$. [S]

(b) $f(x) = (\cosh 2x)^2 - \sin x$; $x_0 = \pi$; $h = 0.05$. [A]

(c) $f(x) = \ln(2x-3) + 5x$; $x = 10$; $h = 1$.

In[483]:=
```
centeredDiff[f_, x0_, h_] := (f[x0 + h] - f[x0 - h]) / (2 h)
results11[g_, x0_, h_, mine_, maxe_, exercise_] :=
 Module[{true = N[g'[x0]], estimate = centeredDiff[g, x0, h], factor = -h^2 / 6},
  Print[exercise];
  TableForm[{{estimate}, {true}, {Abs[true - estimate]},
    {Abs[factor * mine]}, {Abs[factor * maxe]}}, TableHeadings → {{"centered",
      "Mathematica", "abs error", "min error", "max error"}, {"value"}}]
 ]
```

In[485]:=
```
g[x_] := E^x
g'''[x]
{x0, h} = {2, 0.1};
Plot[g'''[x], {x, x0 - h, x0 + h}]
{minerror, maxerror} = {E^(x0 - h), E^(x0 + h)};
results11[g, x0, h, minerror, maxerror, "11(a)"]
```

Out[486]= e^x

Out[488]=



11(a)

Out[490]//TableForm=

	value
centered	7.40138
Mathematica	7.38906
abs error	0.0123213
min error	0.0111432
max error	0.0136103

In[491]:=
```
g[x_] := Cosh[2 x] ^ 2 - Sin[x]
g'''[x]
{x0, h} = {Pi, 0.05};
Plot[g'''[x], {x, x0 - h, x0 + h}]
{minerror, maxerror} = {g'''[x0 - h], g'''[x0 + h]};
results11[g, x0, h, minerror, maxerror, "11(b)"]
```

Out[492]= Cos[x] + 64 Cosh[2 x] Sinh[2 x]

Out[494]=



11(b)

Out[496]//TableForm=

	value
centered	288 668.
Mathematica	286 752.
abs error	1915.5
min error	1565.15
max error	2334.93

In[497]:=
```
g[x_] := Log[2 x - 3] + 5 x
g'''[x]
{x0, h} = {10, 1.0};
Plot[g'''[x], {x, x0 - h, x0 + h}]
{minerror, maxerror} = {g'''[x0 + h], g'''[x0 - h]};
results11[g, x0, h, minerror, maxerror, "11(c)"]
```

Out[498]= $\frac{16}{(-3+2x)^3}$

Out[500]=



11(c)

Out[502]//TableForm=

	value
centered	5.11819
Mathematica	5.11765
abs error	0.00054733
min error	0.000388784
max error	0.000790123

Exercise #18abfg

18. Find the error term for the derivative approximation:

(a) $f'(x_0) \approx \dfrac{f(x_0+2h) - f(x_0)}{2h}$ [A]

(b) $f'(x_0) \approx \dfrac{f(x_0+2h) - f(x_0-h)}{3h}$

(c) $f'(x_0) \approx \dfrac{-3f(x_0) + 4f(x_0+\frac{h}{2}) - f(x_0+h)}{h}$ [S]

(d) $f'(x_0) \approx \dfrac{-13f(x_0-10h) - 12f(x_0+5h) + 25f(x_0+8h)}{270h}$

(e) $f'(x_0) \approx \dfrac{-7f(x_0+h) + 416f(x_0+\frac{1}{2}h) - 2916f(x_0+\frac{1}{3}h) + 5632f(x_0+\frac{1}{4}h) - 3125f(x_0+\frac{1}{5}h)}{12h}$ [A]

(f) $f''(x_0) \approx \dfrac{2f(x_0-h) - 3f(x_0) + f(x_0+2h)}{3h^2}$

(g) $f''(x_0) \approx \dfrac{7f(x_0-5h) - 12f(x_0) + 5f(x_0+7h)}{210h^2}$ [A]

18(a) -- Answer: O(hf''(xi))

In[503]:=
```
Clear[f, x0, h, x]
Series[f[2 h + x0], {h, 0, 3}]
Series[f[h + x0], {h, 0, 3}]
Series[f[x0], {h, 0, 3}]
Series[(f[x0 + 2 h] - f[x0]) / (2 h), {h, 0, 1}]
```

Out[504]= $f[x0] + 2 f'[x0]\, h + 2 f''[x0]\, h^2 + \frac{4}{3} f^{(3)}[x0]\, h^3 + O[h]^4$

Out[505]= $f[x0] + f'[x0]\, h + \frac{1}{2} f''[x0]\, h^2 + \frac{1}{6} f^{(3)}[x0]\, h^3 + O[h]^4$

Out[506]= $f[x0]$

Out[507]= $f'[x0] + f''[x0]\, h + O[h]^2$

18(b)

In[508]:=
```
Clear[f, x0, h, x]
Series[f[2 h + x0], {h, 0, 3}]
Series[f[-h + x0], {h, 0, 3}]
Series[(f[x0 + 2 h] - f[x0 - h]) / (3 h), {h, 0, 1}]
```

Out[509]= $f[x0] + 2 f'[x0]\, h + 2 f''[x0]\, h^2 + \frac{4}{3} f^{(3)}[x0]\, h^3 + O[h]^4$

Out[510]= $f[x0] - f'[x0]\, h + \frac{1}{2} f''[x0]\, h^2 - \frac{1}{6} f^{(3)}[x0]\, h^3 + O[h]^4$

Out[511]= $f'[x0] + \frac{1}{2} f''[x0]\, h + O[h]^2$

18(f)

In[512]:=
```
Clear[f, x0, h, x]
Series[f[2 h + x0], {h, 0, 3}]
Series[f[-h + x0], {h, 0, 3}]
Series[(f[x0 + 2 h] - 3 f[x0] + 2 f[x0 - h]) / (3 h^2), {h, 0, 1}]
```

Out[513]= $f[x0] + 2 f'[x0]\, h + 2 f''[x0]\, h^2 + \frac{4}{3} f^{(3)}[x0]\, h^3 + O[h]^4$

Out[514]= $f[x0] - f'[x0]\, h + \frac{1}{2} f''[x0]\, h^2 - \frac{1}{6} f^{(3)}[x0]\, h^3 + O[h]^4$

Out[515]= $f''[x0] + \frac{1}{3} f^{(3)}[x0]\, h + O[h]^2$

18(g)-- Answer: O(hf’’’(xi))

In[516]:=
```
Clear[f, x0, h, x]
Series[f[7 h + x0], {h, 0, 3}]
Series[f[-5 h + x0], {h, 0, 3}]
Series[(5 f[x0 + 7 h] - 12 f[x0] + 7 f[x0 - 5 h]) / (210 h^2), {h, 0, 1}]
```

Out[517]= $f[x0] + 7 f'[x0]\, h + \frac{49}{2} f''[x0]\, h^2 + \frac{343}{6} f^{(3)}[x0]\, h^3 + O[h]^4$

Out[518]= $f[x0] - 5 f'[x0]\, h + \frac{25}{2} f''[x0]\, h^2 - \frac{125}{6} f^{(3)}[x0]\, h^3 + O[h]^4$

Out[519]= $f''[x0] + \frac{2}{3} f^{(3)}[x0]\, h + O[h]^2$

Exercise #21

21. What can you say about the error in approximating the first derivative of

$$f(x) = -13x^4 + 17x^3 - 15x^2 + 12x - 99$$

using a 5-point formula?

It will have zero error -- the error is proportional to the nth derivative, where n is number of data points. The fifth derivative of f is zero ...

Exercise #22

22. Let $f(x) = 3x^3 - 2x^2 + x$.

(a) Compute the error (not a bound on the error) in estimating $f'(2)$ using the forward difference

$$\frac{f(x_0 + h) - f(x_0)}{h}$$

with $h = 0.1$.

(b) Find $\xi_{0.1}$ as guaranteed by the error term.

In[520]:=
```
f[x_] := 3 x^3 - 2 x^2 + x
h = 0.1;
fp = f'[2]
f''[x]
fd = (f[2 + h] - f[2]) / h
error = Solve[-h / 2 f''[xi] == fp - fd, xi]
```

Out[522]= 29

Out[523]= $-4 + 18\,x$

Out[524]= 30.63

Out[525]= {{xi → 2.03333}}

The error of the forward difference formula is O(hf''(xi)). The second derivative of f(x) is 18x-4.

Exercise #25

25. Refer to the quadrature method

$$\int_{x_0}^{x_0+h} f(x)\,dx = \frac{h}{2}\left[f\left(x_0+\frac{h}{3}\right)+f\left(x_0+\frac{2h}{3}\right)\right]+\frac{h^3}{36}f''(\xi)$$

in all of the following questions. [A]

(a) What is the rate of convergence?

(b) What is the degree of precision?

(c) Use the method to approximate $\int_0^{\pi} \sin x\,dx$.

(d) Find a bound on the error of this approximation.

(e) Compare the bound to the actual error.

(a) rate of convergence is O(h^3)

(b) degree of precision: linear

(c)

In[526]:=
```
f[x_] := Sin[x]
h = Pi;
x0 = 0;
estimate = h / 2.0 (f[x0 + h / 3] + f[x0 + 2 h / 3])
true = NIntegrate[f[x], {x, 0, Pi}]
abserror  =  Abs[true - estimate]
```

Out[529]= 2.7207

Out[530]= 2.

Out[531]= 0.720699

(d) The max size of f''[x]=-sin(x) on [0,Pi] is 1, and the min is 0. So h^3/36 is a bound on abs error:

In[532]:=
```
f''[x]
N[h^3 / 36]
```

Out[532]=
```
-Sin[x]
```

Out[533]=
```
0.861285
```

(e) Our actual error of 0.720699046351326 is below the bound of 0.861285463341662.

Exercise #40

40. The quadrature formula $\int_0^2 f(x)dx = c_0 f(0) + c_1 f(1) + c_2 f(2)$ is exact for all polynomials of degree less than or equal to 2. Determine c_0, c_1, and c_2.

In[534]:=
```
Clear[p0, p1, p2]
x0 = 0;
h = 1;
xpoints = {0, 1, 2};
p0[x_] := 1.0 + x * 0 (* Needed this silly second term to get p0 vectorized....*)
p1[x_] := (x - 1)
p2[x_] := (x - 1) ^2
int0 = Integrate[p0[x], {x, 0, 2}];
int1 = Integrate[p1[x], {x, 0, 2}];
int2 = Integrate[p2[x], {x, 0, 2}];
m = {p0[xpoints], p1[xpoints], p2[xpoints]}; (* without that term,
it just produced "1", rather than {1,1,1}! Yikes....*)
b = {int0, int1, int2};
MatrixForm[m]
MatrixForm[b]
LinearSolve[m, b] // MatrixForm
(* The linear system is simple enough to solve by hand, but I'm lazy!:) *)
```

Out[545]//MatrixForm=

$$\begin{pmatrix} 1. & 1. & 1. \\ -1 & 0 & 1 \\ 1 & 0 & 1 \end{pmatrix}$$

Out[546]//MatrixForm=

$$\begin{pmatrix} 2. \\ 0 \\ \frac{2}{3} \end{pmatrix}$$

Out[547]//MatrixForm=

$$\begin{pmatrix} 0.333333 \\ 1.33333 \\ 0.333333 \end{pmatrix}$$

So the solution is {a,b,c}={1/3,4/3,1/3} -- that’s Simpson’s rule!